*
*
* Moral suasion and the private provision of public goods: Evidence from the COVID-19 pandemic
*
* Bos, B., Drupp, M.A., Meya, J.N., Quaas, M.F. (2020)
*
*


clear all
cls
set more off

* Import data
cd "/Users/bjoern/Desktop/Moral_Suasion_Online_Appendix/"
use "1_Data/MoralSuasion_survey_data.dta", clear 


// Store the location of the results
global results "/Users/bjoern/Desktop/Moral_Suasion_Online_Appendix/4_Tables"


********************************************************************************
* 
* Data preparation
*
*
********************************************************************************

label define moral_treat_label	1 "Conrol" ///
								2 "Deont." /// 
								3 "Conseq."

label value moral_treatment moral_treat_label


// Use a female dummy instead of the gender dummy
gen female = 1 if gender == 2
replace female = 0 if gender == 1
label variable female "Female"


// We drop fast and slow respondents
drop if time_required < 3
drop if time_required > 60


// We drop respondents with the gender "diverse" as there are too few individuals in this group
drop if gender == 3 


// We drop respondents with an invalid answer in the validity check for the moral treatment
tab valid, missing
drop if valid != 1



// Define covariates
global covariates age i.gender i.educ i.hh_inc

// Define outcomes
global outcomes change_contacts_next change_cleaning_hands_next z_support z_change_wrt_gov


********************************************************************************
*
* Statistics mentioned in the text
*
*
********************************************************************************

su protect_me, detail
su protect_ff, detail
su protect_others, detail



********************************************************************************
* 
* Table 1: Main results
*
*
********************************************************************************
eststo clear


** OLS

foreach outcome in $outcomes {

eststo: regress `outcome' i.moral_treatment, robust
	estadd local covariates "No"
eststo: regress `outcome' i.moral_treatment $covariates, robust
	estadd local covariates "Yes"
}

* Save output
esttab using "$results/table_1_OLS.tex", replace ///
	keep(2.moral_treatment 3.moral_treatment) ///
	b(3) ///
	se(3) ///
	star(* 0.1 ** 0.05 *** 0.01) ///
	stats(N, label("Observations") fmt(%9.0fc)) ///
	label ///
	nomtitles ///
	collabels(none) ///
	parentheses ///
	plain ///
	fragment

* With p-values output
esttab using "$results/table_1_OLS_p-values.tex", replace ///
	keep(2.moral_treatment 3.moral_treatment) ///
	mgroups("Planned Cont." "Planned HC" "\makecell{Support for\\ gov. reg.\\ (z-score)}" "\makecell{Change cont.\\ wrt. gov. reg.\\ (z-score)}", pattern(1 0 1 0 1 0 1 0) ///
		prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span})) ///
	title("Main results (OLS with p-values)") ///
	posthead("`numbers'") ///
	nomtitles ///
	booktabs ///
	nobaselevels ///
	nonumbers ///
	width(\textwidth) ///
	b(3) ///
	p(3) ///
	scalars("covariates Covariates") ///
	obslast ///
	stats(N, label("Observations") fmt(%9.0fc)) ///
	star(* 0.1 ** 0.05 *** 0.01) ///
	label

* Drop regression results
eststo clear
	


** Tobit
// Change in contacts
eststo: tobit change_contacts_next i.moral_treatment, ll(1) ul(15) vce(robust)
	estadd local covariates "No"
eststo: tobit change_contacts_next i.moral_treatment $covariates, ll(1) ul(15)  vce(robust)
	estadd local covariates "Yes"

// Change in hand cleaning effort
eststo: tobit change_cleaning_hands_next i.moral_treatment, ll(1) ul(15)  vce(robust)
	estadd local covariates "No"
eststo: tobit change_cleaning_hands_next i.moral_treatment $covariates, ll(1) ul(15)  vce(robust)
	estadd local covariates "Yes"

// Support for gov. regulation
quietly su z_support // This yields the min and max of the z-score
eststo: tobit z_support i.moral_treatment, ll(r(min)) ul(r(max))  vce(robust)
	estadd local covariates "No"

quietly su z_support // This yields the min and max of the z-score
eststo: tobit z_support i.moral_treatment $covariates, ll(r(min)) ul(r(max))  vce(robust)
	estadd local covariates "Yes"

// Change in contacts wrt. gov. regulation
quietly su z_change_wrt_gov // This yields the min and max of the z-score
eststo: tobit z_change_wrt_gov i.moral_treatment, ll(r(min)) ul(r(max))  vce(robust)
	estadd local covariates "No"

quietly su z_change_wrt_gov // This yields the min and max of the z-score
eststo: tobit z_change_wrt_gov i.moral_treatment $covariates, ll(r(min)) ul(r(max))  vce(robust)
	estadd local covariates "Yes"
	

* Save output
esttab using "$results/table_1_Tobit.tex", replace ///
	keep(2.moral_treatment 3.moral_treatment) ///
	b(3) ///
	se(3) ///
	star(* 0.1 ** 0.05 *** 0.01) ///
	stats(N, label("Observations") fmt(%9.0fc)) ///
	label ///
	nomtitles ///
	collabels(none) ///
	parentheses ///
	plain ///
	fragment

* Drop regression results
eststo clear



********************************************************************************
*
* Table A1: Descriptive statistics of relevant survey respondenses
*
*
********************************************************************************

eststo clear

* Generate dummies for factor variables
*	 (Reason: estpost does not work with factor variables)

label variable change_contacts_next "Change in planned contacts"
label variable change_cleaning_hands_next "Change in planned hand cleaning effort"


* Grouping of subjects wrt their behaviour compared the the
gen group_support = 1 if perception_gov_reg == 6 // Are appropriate
replace group_support = 2 if perception_gov_reg < 6 // Are too much
replace group_support = 3 if perception_gov_reg > 6 // Are too little

label variable group_support "Support for gov. regulation"

label define labelgroup_sup	1 "Are appropriate" ///
							2 "Are too much" /// 
							3 "Are too little"
							
label value group_support labelgroup_sup

tab group_support, gen(group_supportx)

label variable group_supportx1 "\hspace{0.2cm} Are appropriate"
label variable group_supportx2 "\hspace{0.2cm} Are too much"
label variable group_supportx3 "\hspace{0.2cm} Are too little"



* Grouping of subjects wrt their behaviour compared the the
gen group_reg = 1 if change_con_wrt_gov_reg == 6 // According to regulations
replace group_reg = 2 if change_con_wrt_gov_reg < 6 // Less than required
replace group_reg = 3 if change_con_wrt_gov_reg > 6 // More than required

label variable group_reg "Contacts wrt. regulation"

label define labelgroup_reg	1 "According to regulations" ///
							2 "Less than required" /// 
							3 "More than required"
							
label value group_reg labelgroup_reg

tab group_reg, gen(group_regx)

label variable group_regx1 "\hspace{0.2cm} According to regulations"
label variable group_regx2 "\hspace{0.2cm} Less than required"
label variable group_regx3 "\hspace{0.2cm} More than required"



global desc_var	change_contacts_next change_cleaning_hands_next ///
				group_supportx2 group_supportx1 group_supportx3 ///
				group_regx2 group_regx1 group_regx3
				
				
**** Summary statistics overall and by population group

eststo: estpost summarize $desc_var
eststo: estpost summarize $desc_var if pop_group == 1 // Young men
eststo: estpost summarize $desc_var if pop_group == 2 // Young women
eststo: estpost summarize $desc_var if pop_group == 3 // Old men
eststo: estpost summarize $desc_var if pop_group == 4 // Old women


esttab using "$results/table_A1_desc_stat.tex", replace ///
	refcat(group_supportx2 "\emph{Support for gov. regulation (in \%)}" ///
		   group_regx2 "\emph{Change wrt. regulation (in \%)}", nolabel) ///
	title("Descriptive statistics of relevant survey responses. \label{tab:desc-stat-survey}") ///
	mtitle("\makecell{ \\ \\}" ///
			"\makecell{Young \\ men}" ///
			"\makecell{Young \\ women}" ///
			"\makecell{Old \\ men}" ///
			"\makecell{Old \\ women}") ///
	mgroups("All" "Population Group", pattern(1 1 0 0 0) ///
			prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span})) ///
	collabels(none) ///
	cells(mean(fmt(2)) sd(par fmt(2))) ///
	label ///
	nonum ///
	booktabs ///
	addnotes("\emph{Notes:} Table shows mean values and standard deviations in parentheses.")


eststo clear


**** t-tests for the differences
// Generate dummies for our population groups
gen group_1 = 1 if pop_group == 1 // young, Male
gen group_2 = 1 if pop_group == 2 // young, Female
gen group_3 = 1 if pop_group == 3 // old, Male
gen group_4 = 1 if pop_group == 4 // old, Female

replace group_1 = 0 if missing(group_1)
replace group_2 = 0 if missing(group_2)
replace group_3 = 0 if missing(group_3)
replace group_4 = 0 if missing(group_4)

eststo clear

eststo: estpost ttest $desc_var, by(group_1)
eststo: estpost ttest $desc_var, by(group_2)
eststo: estpost ttest $desc_var, by(group_3)
eststo: estpost ttest $desc_var, by(group_4)


esttab using "$results/table_A1_ttests.tex", replace ///
	cells("mu_2(fmt(2)) _star" "se(par)") ///
	wide ///
	refcat(group_supportx2 "\emph{Support for gov. regulation (in \%)}" ///
		   group_regx2 "\emph{Change wrt. regulation (in \%)}", nolabel) ///
	 title("Descripte statistiscs: t-tests for the differences") ///
	 mtitle("\makecell{Young \\ men \\ (1)}" ///
			"\makecell{Young \\ women \\ (2)}" ///
			"\makecell{Old \\ men \\ (3)}" ///
			"\makecell{Old \\ women \\ (4)}") ///
	 collabels(none) ///
	 noobs ///
	 label ///
	 nonum ///
	 booktabs ///
	 addnotes("\parbox{12cm}{\emph{Notes:} Table shows mean values per subgroup and standard errors in parentheses. Stars indicate the significance of the mean values to the average mean values of the other groups (t-tests). \sym{*} \(p<0.1\), \sym{**} \(p<0.05\), \sym{***} \(p<0.01\)}")


eststo clear



********************************************************************************
*
* Table A2: Balance tests
*
*
********************************************************************************

tab hh_inc, gen(hh_incx)
tab educ, gen(educx)

label variable educx1 "\hspace{0.2cm} University degree"
label variable educx2 "\hspace{0.2cm} A-levels / vocational training"
label variable educx3 "\hspace{0.2cm} Secondary school"
label variable educx4 "\hspace{0.2cm} Secondary general school"
label variable educx5 "\hspace{0.2cm} No degree"

label variable hh_incx1 "\hspace{0.2cm} $<$ 1,500"
label variable hh_incx2 "\hspace{0.2cm} 1,500 -- 3,000"
label variable hh_incx3 "\hspace{0.2cm} 3,000 -- 4,000"
label variable hh_incx4 "\hspace{0.2cm} $\geq$ 4,000"


global balance_var	age female hh_incx* educx*


eststo clear

// Mean + SD of control group
eststo: estpost summarize $balance_var if moral_treatment == 1 // Control
eststo: estpost summarize $balance_var if moral_treatment == 2 // Deont
eststo: estpost summarize $balance_var if moral_treatment == 3 // Conseq.

esttab using "$results/table_A2_balance.tex", replace ///
	refcat(hh_incx1 "\emph{Household income}" ///
		   educx1 "\emph{Education}", nolabel) ///
	title("Balance test \label{tab:balance-test}") ///
	mtitle("\makecell{Control}" ///
		   "\makecell{Deont.}" ///
		   "\makecell{Conseq.}") ///
	collabels(none) ///
	cells(mean(fmt(2)) sd(par fmt(2))) ///
	label ///
	nonum ///
	booktabs ///
	addnotes("\emph{Notes:} Table shows mean values and standard deviations in parentheses.")

eststo clear


// ttests for deont. and conseq. treatment groups
eststo: estpost ttest $balance_var if moral_treatment != 3, by(moral_treatment) // Control (1) vs. Deont (2)
eststo: estpost ttest $balance_var if moral_treatment != 2, by(moral_treatment) // Control (1) vs. Conseq. (3)

esttab using "$results/table_A2_ttest.tex", replace ///
	refcat(hh_incx1 "\emph{Household income}" ///
		   educx1 "\emph{Education}", nolabel) ///
	cells("mu_2(fmt(2)) _star" "se(par)") ///
	wide ///
	title("Balance test (Part 2)") ///
	mtitle("\makecell{Deont.}" ///
		   "\makecell{Conseq.}") ///
	collabels(none) ///
	noobs ///
	label ///
	nonum ///
	booktabs ///
	addnotes("\parbox{12cm}{\emph{Notes:} Table shows mean values per subgroup and standard errors in parentheses. Stars indicate the significance of the mean values to the average mean values of the other groups (t-tests). \sym{*} \(p<0.1\), \sym{**} \(p<0.05\), \sym{***} \(p<0.01\)}")

	 
eststo clear



********************************************************************************
* 
* A3: Heterogeneous treatment effects: young vs. old & male vs. female
*
*
********************************************************************************
eststo clear

** Young respondents

foreach outcome in $outcomes {

eststo: regress `outcome' i.moral_treatment if age < 60, robust
	estadd local covariates "No"
eststo: regress `outcome' i.moral_treatment $covariates prob_get_ill if age < 60, robust
	estadd local covariates "Yes"

}

* Save output
esttab using "$results/table_A3_young.tex", replace ///
	keep(2.moral_treatment 3.moral_treatment) ///
	nomtitles ///
	booktabs ///
	nobaselevels ///
	nonumbers ///
	width(\textwidth) ///
	b(3) ///
	se(3) ///
	scalars("covariates Covariates") ///
	obslast ///
	stats(N, label("Observations") fmt(%9.0fc)) ///
	star(* 0.1 ** 0.05 *** 0.01) ///
	label ///
	fragment ///
	parentheses ///
	plain ///
	eqlabels(none) ///
	collabels(none)

eststo clear


** Old respondents

foreach outcome in $outcomes {

eststo: regress `outcome' i.moral_treatment if age >= 60, robust
	estadd local covariates "No"
eststo: regress `outcome' i.moral_treatment $covariates prob_get_ill if age >= 60, robust
	estadd local covariates "Yes"

}
	
* Save output
esttab using "$results/table_A3_old.tex", replace ///
	keep(2.moral_treatment 3.moral_treatment) ///
	nomtitles ///
	booktabs ///
	nobaselevels ///
	nonumbers ///
	width(\textwidth) ///
	b(3) ///
	se(3) ///
	scalars("covariates Covariates") ///
	obslast ///
	stats(N, label("Observations") fmt(%9.0fc)) ///	
	star(* 0.1 ** 0.05 *** 0.01) ///
	label ///
	fragment ///
	parentheses ///
	plain ///
	eqlabels(none) ///
	collabels(none)


eststo clear


** Male respondents

foreach outcome in $outcomes {

eststo: regress `outcome' i.moral_treatment if gender == 1, robust
	estadd local covariates "No"
eststo: regress `outcome' i.moral_treatment $covariates if gender == 1, robust
	estadd local covariates "Yes"

}

* Save output
esttab using "$results/table_A3_male.tex", replace ///
	keep(2.moral_treatment 3.moral_treatment) ///
	nomtitles ///
	booktabs ///
	nobaselevels ///
	nonumbers ///
	width(\textwidth) ///
	b(3) ///
	se(3) ///
	scalars("covariates Covariates") ///
	obslast ///
	stats(N, label("Observations") fmt(%9.0fc)) ///
	star(* 0.1 ** 0.05 *** 0.01) ///
	label ///
	fragment ///
	parentheses ///
	plain ///
	eqlabels(none) ///
	collabels(none)

eststo clear


** Female respondents

foreach outcome in $outcomes {

eststo: regress `outcome' i.moral_treatment if gender == 2, robust
	estadd local covariates "No"
eststo: regress `outcome' i.moral_treatment $covariates if gender == 2, robust
	estadd local covariates "Yes"

}

* Save output
esttab using "$results/table_A3_female.tex", replace ///
	keep(2.moral_treatment 3.moral_treatment) ///
	nomtitles ///
	booktabs ///
	nobaselevels ///
	nonumbers ///
	width(\textwidth) ///
	b(3) ///
	se(3) ///
	scalars("covariates Covariates") ///
	obslast ///
	stats(N, label("Observations") fmt(%9.0fc)) ///
	star(* 0.1 ** 0.05 *** 0.01) ///
	label ///
	fragment ///
	parentheses ///
	plain ///
	eqlabels(none) ///
	collabels(none)

eststo clear


********************************************************************************
* 
* Table A4: Heterogeneous treatment effects: 3 age groups
*
*
********************************************************************************


gen age_class = 1 if age < 30 & !missing(age)
replace age_class = 2 if age >= 30 & age <= 65 & !missing(age)
replace age_class = 3 if age > 65 & !missing(age)

tab age_class


foreach age_group in 1 2 3 {

	foreach outcome in $outcomes {

	eststo: regress `outcome' i.moral_treatment if age_class == `age_group', robust
		estadd local covariates "No"
	eststo: regress `outcome' i.moral_treatment $covariates prob_get_ill if age_class == `age_group', robust
		estadd local covariates "Yes"
	}

* Save output
esttab using "$results/table_A4_age_group_`age_group'.tex", replace ///
	keep(2.moral_treatment 3.moral_treatment) ///
	nomtitles ///
	booktabs ///
	nobaselevels ///
	nonumbers ///
	width(\textwidth) ///
	b(3) ///
	se(3) ///
	scalars("covariates Covariates") ///
	stats(N, label("Observations") fmt(%9.0fc)) ///
	obslast ///
	star(* 0.1 ** 0.05 *** 0.01) ///
	label ///
	fragment ///
	parentheses ///
	plain ///
	eqlabels(none) ///
	collabels(none)

eststo clear
}

	
********************************************************************************
* 
* Table A5: Heterogeneous treatment effects: altruism groups
*
*
********************************************************************************



** Define altruism groups
egen protect_me_median = median(protect_me)
egen protect_others_median = median(protect_others)

gen altruism_class = 1 if protect_me == protect_me_median & !missing(protect_me)
replace altruism_class = 2 if protect_me < protect_me_median & !missing(protect_me)
replace altruism_class = 3 if protect_me > protect_me_median & !missing(protect_me)

label define label_altruism_	1 "\hspace{0.2cm} Median" ///
								2 "\hspace{0.2cm} Lower than median, i.e. high altruism" /// 
								3 "\hspace{0.2cm} Higher than median, i.e., low altruism"
								
label value altruism_class label_altruism_
tab altruism_class


eststo clear

// Altruism based on share put on me
foreach group_code in 1 2 3 {

	foreach outcome in $outcomes {
		
	eststo: regress `outcome' i.moral_treatment if altruism_class == `group_code', robust
		estadd local covariates "No"
	eststo: regress `outcome' i.moral_treatment $covariates if altruism_class == `group_code', robust
		estadd local covariates "Yes"

	}

* Save output
esttab using "$results/table_A5_altruism_group_`group_code'.tex", replace ///
	keep(2.moral_treatment 3.moral_treatment) ///
	nomtitles ///
	booktabs ///
	nobaselevels ///
	nonumbers ///
	width(\textwidth) ///
	b(3) ///
	se(3) ///
	scalars("covariates Covariates") ///
	stats(N, label("Observations") fmt(%9.0fc)) ///
	obslast ///
	star(* 0.1 ** 0.05 *** 0.01) ///
	label ///
	fragment ///
	parentheses ///
	plain ///
	eqlabels(none) ///
	collabels(none)	
	
eststo clear
}


eststo clear
// Altruism based on share put on others

foreach outcome in $outcomes {
	
eststo: regress `outcome' i.moral_treatment if protect_others > protect_others_median & !missing(protect_others), robust
	estadd local covariates "No"
eststo: regress `outcome' i.moral_treatment $covariates if protect_others > protect_others_median & !missing(protect_others), robust
	estadd local covariates "Yes"

}

* Save output
esttab using "$results/table_A5_altruism_group_4.tex", replace ///
	keep(2.moral_treatment 3.moral_treatment) ///
	nomtitles ///
	booktabs ///
	nobaselevels ///
	nonumbers ///
	width(\textwidth) ///
	b(3) ///
	se(3) ///
	scalars("covariates Covariates") ///
	stats(N, label("Observations") fmt(%9.0fc)) ///
	obslast ///
	star(* 0.1 ** 0.05 *** 0.01) ///
	label ///
	fragment ///
	parentheses ///
	plain ///
	eqlabels(none) ///
	collabels(none)

eststo clear


********************************************************************************
* 
* Table A7: Multiple Hypothesis Testing
*
* (run before A6, as A6 loads the data again)
********************************************************************************


** Using mhtexp from  John List, Azeem Shaikh and Yang Xu (2016)
// ssc install mhtexp

// Use a numeric treatment variable being zero for the control group
gen treat = 0 if moral_treatment == 1
replace treat = 1 if moral_treatment == 2
replace treat = 2 if moral_treatment == 3

mhtexp change_contacts_next change_cleaning_hands_next z_support z_change_wrt_gov, treatment(treat) bootstrap(3000)



********************************************************************************
*
* Table A6: Robustness checks
*
*
********************************************************************************

// Import the data but we do not exlude fast and slow clickers:
use "1_Data/1_Wave/2_processed/MoralSuasion_survey_data.dta", clear

gen ban = 1 if submit_timestamp > clock("2020-03-22 19:00:00", "YMDhms")
replace ban = 0 if missing(ban)
label variable ban "Post ban announcement"


label define moral_treat_label	1 "Conrol" ///
								2 "Deont." /// 
								3 "Conseq."

label value moral_treatment moral_treat_label


// Use a female dummy instead of the gender dummy
gen female = 1 if gender == 2
replace female = 0 if gender == 1
label variable female "Female"


// Reweight the control group to match the treatment groups
gen deont_treat = 1 if moral_treatment == 2
replace deont_treat = 0 if moral_treatment == 1

gen conseq_treat = 1 if moral_treatment == 3
replace conseq_treat = 0 if moral_treatment == 1

tab educ, gen(educx)
tab hh_inc, gen(hh_incx)

ebalance deont_treat age gender educx1 educx2 educx3 educx4 educx5 ///
		hh_incx1 hh_incx2 hh_incx3, targets(1) gen(deont_weights)

ebalance conseq_treat age gender educx1 educx2 educx3 educx4 educx5 ///
		hh_incx1 hh_incx2 hh_incx3, targets(1) gen(conseq_weights)


		
// Loop over each outcome 
foreach outcome in $outcomes {

eststo clear


eststo: regress `outcome' i.moral_treatment $covariates if time_required >= 3 & time_required <= 60 & valid == 1, robust
	estadd local covariates "Yes"
	estadd local valid "Yes"
	estadd local ban "No"
	estadd local clickers "No"
	estadd local weighted "No"
	
eststo: regress `outcome' i.moral_treatment i.ban $covariates if time_required >= 3 & time_required <= 60 & valid == 1, robust
	estadd local covariates "Yes"
	estadd local valid "Yes"
	estadd local ban "Yes"
	estadd local clickers "No"
	estadd local weighted "No"
		
eststo: regress `outcome' i.moral_treatment $covariates if valid == 1, robust
	estadd local covariates "Yes"
	estadd local valid "Yes"
	estadd local ban "No"
	estadd local clickers "Yes"
	estadd local weighted "No"
	
eststo: regress `outcome' i.moral_treatment $covariates if time_required >= 3 & time_required <= 60, robust
	estadd local covariates "Yes"
	estadd local valid "No"
	estadd local ban "No"
	estadd local clickers "No"
	estadd local weighted "No"
	
eststo: regress `outcome' i.moral_treatment $covariates [aw = deont_weights] if time_required >= 3 & time_required <= 60 & valid == 1 & moral_treatment != 3, robust
	estadd local covariates "Yes"
	estadd local valid "Yes"
	estadd local ban "No"
	estadd local clickers "No"
	estadd local weighted "Yes"
	
eststo: regress `outcome' i.moral_treatment $covariates [aw = conseq_weights] if time_required >= 3 & time_required <= 60 & valid == 1 & moral_treatment != 2, robust
	estadd local covariates "Yes"
	estadd local valid "Yes"
	estadd local ban "No"
	estadd local clickers "No"
	estadd local weighted "Yes"

	
* Save output
esttab using "$results/table_A6_`outcome'.tex", replace ///
	keep(2.moral_treatment 3.moral_treatment) ///
	nomtitles ///
	booktabs ///
	nobaselevels ///
	nonumbers ///
	b(3) ///
	se(3) ///
	scalars("covariates Covariates" "valid Valid" "ban Ban-announcement" "clickers Clickers" "weighted Weighted") ///
	stats(N, label("Observations") fmt(%9.0fc)) ///
	obslast ///
	star(* 0.1 ** 0.05 *** 0.01) ///
	label ///
	fragment ///
	plain ///
	eqlabels(none) ///
	collabels(none) ///
	parentheses

}

